

# **Introducción a la Programación usando Pascal como primer lenguaje**

## **Capítulo 2 Elementos básicos**

*Francisco J. Ballesteros*

## ***Elementos básicos***

### **2.1. ¿Por dónde empezamos?**

En el capítulo anterior hemos tomado contacto con la programación y con el lenguaje Pascal. Por el momento, una buena forma de proceder es tomar un programa ya hecho (como por ejemplo el “Hola ” del capítulo anterior) y cambiar sólo las partes de ese programa que nos interesen. De ese modo podremos ejecutar nuestros propios programas sin necesidad de, por el momento, saber cómo escribirlos enteros por nuestros propios medios.

Recuerda que programar es como cocinar (o como conducir). Sólo se puede aprender a base de práctica y a base de ver cómo practican otros. No es posible aprenderlo realmente sólo con leer libros (aunque eso ayuda bastante). Si no tienes el compilador de Pascal cerca, descárgalo de

<http://freepascal.org>


salvo para probar por ti mismo cada una de las cosas que veas durante este curso. La única forma de aprender a usarlas es usándolas. El lenguaje Pascal permite básicamente manipular datos. Eso sí, estos datos son entidades abstractas y están alejadas de lo que en realidad entiende el ordenador. Programar en Pascal consiste en aprender a definir y manipular estas entidades abstractas, lo cual es más sencillo de lo que parece.

¿Cómo es el mundo para Pascal? En realidad es tan sencillo como muestra la figura 1. Para Pascal, en el mundo existen entidades (cosas) que pueden ser números reales, números enteros, caracteres o valores de verdad. ¡Y no hay nada más! (por el momento).

Esto quiere decir que en Pascal tenemos estos cuatro conjuntos diferentes (números reales, números enteros, caracteres y valores de verdad). Cada uno de ellos consta de elementos homogéneos entre sí. El conjunto de los caracteres contiene, sorprendentemente, caracteres; igual sucede con el conjunto de los números reales, que contiene números reales.

Cada conjunto tiene unas operaciones bien definidas para trabajar sobre sus elementos. Por ejemplo, podemos sumar números enteros entre sí para obtener otros números enteros. Igualmente, podemos sumar números reales entre sí para obtener otros números reales. Pero, en general, *no podemos mezclar números enteros con números reales*.

Quizá sorprenda esto, pero así son las cosas. Si queremos sumar un número entero a un número real tendremos que conseguir un número real que tenga el mismo valor que el número entero, y luego sumarlo. Por fortuna, en este caso, Pascal ayudará convirtiendo enteros a reales en ciertas ocasiones para evitarnos



**Fig. 1.** El mundo según Pascal: existen enteros, caracteres, reales y valores de verdad.

molestias. No obstante, veremos muchas veces en que nadie nos ayuda y vemos la cruda realidad: los enteros y los reales no tienen nada que ver en Pascal (y en muchos otros lenguajes).

Por ejemplo, si calculamos el módulo (resto de la división en positivos) de dos enteros, tenemos que utilizar dos enteros como valores de partida y obtenemos un entero como valor de resultado. No podemos utilizar un número entero y un real. Pascal no sabe combinarlos. La razón por la que esto es así es intentar evitar que, por accidente, hagamos cosas tales como sumar peras con manzanas; o contamos peras o contamos manzanas o contamos piezas de fruta. Pero no podemos mezclarlo todo a voluntad.

A cada conjunto se le denomina **tipo de datos** y, naturalmente, a los elementos de cada conjunto se les denomina **datos**. Resumiendo:

- Un tipo de datos es un conjunto de elementos con unas operaciones bien definidas para datos pertenecientes a ese tipo.
- No es posible combinar datos de distinto tipo entre sí.

Debido a esto se dice que Pascal es un lenguaje **fuertemente tipado**. Hay otros lenguajes de programación que permiten combinar entre sí elementos de distintos

tipos, pero esto da lugar a muchos errores a la hora de programar.

Pascal utiliza la palabra reservada “`integer`” para denominar al tipo de datos (conjunto) de los números enteros. El conjunto de los reales está representado en Pascal por el tipo de datos `real`. De igual manera, tenemos el conjunto de los valores de verdad, “cierto” y “falso”, representado en Pascal por el tipo de datos `boolean`. Este último sólo tiene dos elementos: `True` representa en Pascal al valor de verdad “cierto” y `False` representa al valor de verdad “falso”. Tenemos también un tipo de datos para el conjunto de caracteres disponible, llamado `char`.

Así, el literal 3 pertenece al tipo de datos `integer` dado que es un número entero y sólo puede combinarse normalmente con otros elementos también del tipo `integer`. La siguiente expresión en Pascal intenta calcular la división entera entre 5 y 2:

```
5 div 2.0
```

Un entero sólo puede dividirse con otro entero; nunca con un número real. El punto entre el dos y el cero del segundo operando hace que ese literal sea un número real. Por tanto, dicha expresión es incorrecta y dará lugar a un error durante la compilación del programa que la incluya. Lo hemos intentado (cambiando un poco el programa de saludo a ) y esto es lo que ha pasado:

```
unix% fpc malo.p
```

```
malo.p:13: malo.p(3,13) Error: Operator is not overloaded
```

```
unix%
```

El compilador ha escrito un mensaje para informar de un error, intentando describirlo de la forma más precisa que ha podido, y no ha hecho nada más. Esto es, **no ha compilado el programa**. ¡No tenemos un fichero ejecutable! El programa contenía la siguiente sentencia (hemos cambiado la línea que llama a `writeln` para que imprima la división):

```
writeln(5 div 2.0);
```

y esta sentencia no compila, dado que intenta mezclar elementos de distintos tipos.

La razón es que debe existir **compatibilidad de tipos** entre los elementos que aparecen en una expresión. Esto es, podemos escribir

```
5 div 2
```

y también

```
5.0 / 2.0
```

pero (en general) no podemos mezclar valores de tipo `integer` y valores de tipo `real` como en la expresión incorrecta mostrada arriba.

Los literales de tipo `real` pueden también escribirse en notación exponencial, expresados como un número multiplicado por 10 elevado a una potencia dada. Por ejemplo,

```
3.2E-3
```

es en realidad  $3.2 \cdot 10^{-3}$ , esto es: 0.0032.

En aquellas ocasiones en que necesitamos sumar un entero a un número real, Pascal nos ayuda realizando gratis una **conversión de tipo** sobre el entero para conseguir un valor real equivalente. Por ejemplo, si utilizamos

```
3.0 + 2
```

Pascal se comporta como si escribimos

```
3.0 + real(2)
```

con lo que la operación de suma en realidad está sumando dos números reales. Ambos operandos, y el resultado, son de tipo `real` y Pascal sigue feliz sin tener que mezclar elementos de tipos distintos.

Las conversiones de tipo normalmente no son automáticas y no siempre se permiten. Por ejemplo, no podemos convertir un carácter a un número real, igual que no tendría sentido convertir una pera en una manzana o el plomo en oro (bueno, esto último tal vez sí).

## 2.2. Operaciones

Como estamos viendo, cada tipo de datos tiene un nombre, por ejemplo `integer`, y define un conjunto de elementos de dicho tipo. Además, cada tipo de datos tiene unas operaciones básicas que nos permiten manipular datos de dicho tipo. El reflejo en Pascal de estas operaciones básicas (que son en realidad operaciones matemáticas) son los llamados **operadores**. Por ejemplo, “+” es un operador para el tipo `integer` y podemos utilizarlo para sumar números enteros.

Un operador permite operar sobre uno o más elementos de un tipo de datos determinado. A estos elementos se les denomina **operandos**. Algunos operadores se escriben entre los operandos (por ejemplo, la suma) y se les denomina **infijos**. Otros operadores adoptan otras formas. Por ejemplo, el cambio de signo se escribe con un “-” escrito antes del objeto sobre el que opera; por eso se denomina operador **prefijo**. Otros operadores adoptan el aspecto de funciones, como por ejemplo el operador de conversión de tipo que hemos visto antes para convertir un 2 en un valor de tipo `real`.

Los tipos de datos numéricos admiten los siguientes operadores:

- Suma: `3 + 4`.

Tanto para enteros como para reales. Si se combinan enteros y reales, Pascal convierte el entero a un real antes de efectuar la suma, por lo que la suma es de datos de tipo `real`.

- Resta: `4.0 - 7.2`.

Lo dicho para la suma es aplicable a la resta.

- Multiplicación: `2 * 2`.

Lo dicho para la suma es aplicable a la multiplicación.

- División entera: `7 div 3`.

Ambos operandos han de ser enteros, y el resultado es un número entero

correspondiente a la división entera.

- Módulo: `7 mod 3`.

Ambos operandos han de ser enteros, y el resultado es un número entero correspondiente al módulo (resto de la división entera si son positivos).

- División real: `7 / 3`.

Lo dicho para la suma es aplicable a la división real. En el ejemplo de este caso, la división es exactamente la que habría sucedido utilizando `7.0 / 3.0`.

- Exponenciación: `2 ** 3`. (Esto es,  $2^3$ ).

Sólo disponible para números enteros, y sólo si se utiliza la librería o unidad `math`, como verás en futuros ejemplos.

- Cambio de signo: `- 5`

Tanto para enteros como para reales.

Cada operador sabe operar sólo con un tipo de datos. Esto es así incluso si operadores para distintos tipos de datos se llaman igual. Por ejemplo, la multiplicación “`*`” opera o bien con enteros o bien con reales. Si un operando es entero y el otro real, Pascal hace una conversión del entero a un real para facilitarnos las cosas.

La lista de operadores la hemos incluido aquí más como una referencia que como otra cosa. Podemos olvidarnos de ella por ahora. A medida que los uses los recordarás de forma natural.

Otras muchas funciones están disponibles para operar con valores numéricos gracias a la librería que ofrece el compilador, aunque técnicamente no son parte del lenguaje. Por ejemplo, `trunc(4.2)` es una función que devuelve la parte entera de un número real (consiguiendo el efecto inverso de una conversión de tipo entero a real); `sqr(4.2)` eleva un número real (4.2 en este caso) al cuadrado; `sqrt(4.2)` obtiene la raíz cuadrada de un número real (de 4.2 en este caso); etc.

## 2.3. Expresiones

Empleando los operadores y funciones que hemos descrito es posible escribir expresiones más complejas que calculen valores de forma similar a como hacemos en matemáticas. El problema de la escritura de expresiones radica en que para Pascal el programa es simplemente una única secuencia de caracteres. Así es como ve Pascal el programa para saludar la número :

```
{saludar al numero pi} program holapi; const Pi: real = 3.14; begin  
  writeln ('hola ', Pi:0:2, '!'); end.
```

Lo ve como si todas las líneas del fichero estuvieran escritas en una única línea extremadamente larga. Esto quiere decir que no podemos escribir un 3 sobre el signo de dividir y una suma bajo el mismo para expresar que el denominador es una suma. Por ejemplo, para dividir 3.0 por 2.0 tenemos que escribir `3.0/2.0` pero no podemos

escribir  $\frac{3}{2}$ . Igualmente, para sumar los números del 1 al 5 y dividir el resultado por 3 no podemos utilizar

$$\frac{1 + 2 + 3 + 4 + 5}{3}$$

En su lugar, es preciso escribir:  $(1 + 2 + 3 + 4 + 5) / 3$  (si queremos emplear reales). Los paréntesis son necesarios para agrupar la expresión que suma los números del uno al cinco, de tal forma todas las sumas estén en el numerador. Lo que pasa es que los operadores se evalúan empleando un orden determinado. Por eso se dice que algunos tienen **precedencia** sobre otros (que el lenguaje los evalúa antes que estos otros). Por ejemplo, la división y la multiplicación tienen precedencia sobre las sumas y restas. Esto es, las divisiones y las multiplicaciones en una expresión se calculan antes que las sumas y las restas. En el ejemplo anterior, de no utilizar los paréntesis estaríamos calculando en realidad:

$$1 + 2 + 3 + 4 + 5 / 3$$

Esto es,

$$1 + 2 + 3 + 4 + (5 / 3)$$

O lo que es lo mismo:

$$1 + 2 + 3 + 4 + \frac{5}{3}$$

La exponenciación tiene precedencia sobre la multiplicación y la división. Por tanto,

$$2 * 3 ** 2$$

está calculando

$$2 * (3 ** 2)$$

y no

$$(2 * 3) ** 2$$

Cuando existan dudas sobre la precedencia conviene utilizar paréntesis para agrupar las expresiones según deseemos.

Una nota importante es que *se debe utilizar el espacio en blanco y los signos de puntuación (paréntesis en este caso) para hacer más legible la expresión*. Como puede verse en las expresiones de ejemplo, los espacios están escritos de tal modo que resulta más fácil leerlas. Normalmente, se escriben antes y después del operador, pero nunca después de un paréntesis abierto o antes de uno cerrado.

## 2.4. Otros tipos de datos

El tipo de datos `char` representa un carácter del juego de caracteres disponibles. Estos son algunos ejemplos de literales de tipo carácter:

`'A'`      `'0'`      `' '`

El primero representa a la letra “A”, el segundo al dígito “0” y el último al espacio en blanco. Todos ellos son de tipo `char`. Es importante ver que

```
'0'
```

es un carácter pero en cambio

```
0
```

es un entero. El primero se usa al manipular texto y el segundo al manipular números. ¡No tienen que ver entre sí! De hecho... ¡No pueden operarse entre sí dado que son de distinto tipo!

Un carácter se almacena en el ordenador como un número cuyo valor representa el carácter en cuestión. Inicialmente se utilizaba el código **ASCII** para representar los caracteres. Hoy día son populares otros códigos como **UTF-8**. Una codificación de caracteres no es más que una tabla en la que se asigna un valor numérico a cada carácter que se quiera representar (como ya sabemos, los ordenadores sólo saben manejar números). En la mayoría de los casos, las letras sin acentuar, dígitos y signos comunes de puntuación empleados en Inglés están disponibles en el juego de caracteres ASCII. Suele ser buena idea *no* emplear acentos en los identificadores que empleamos en los programas por esta razón. Aunque muchos lenguajes lo permiten, no es así en otros y no todo el software de ordenador (editores por ejemplo) se comporta igual con caracteres acentuados.

En ocasiones, es útil saber qué posición ocupa un carácter en la tabla ASCII. En Pascal podemos obtener esta correspondencia mediante la función `ord`, que devuelve el ordinal de un carácter o su posición en el código. Por ejemplo, para saber la posición en la tabla del carácter

```
'z'
```

podemos usar la expresión:

```
ord('z')
```

Dicha expresión tiene como valor 90 dado que el carácter indicado ocupa esa posición en el código ASCII. Para conseguir lo inverso, esto es, conseguir el carácter que corresponde a una posición en la tabla, podemos utilizar la función `chr`. Por ejemplo, esta expresión

```
chr(90)
```

tiene como valor

```
'z'
```

de tipo `char`, dado que ese carácter tiene la posición 90 en el código de caracteres.

Otro tipo de datos importante es el llamado booleano, `boolean` en Pascal. Este tipo representa valores de verdad: “Cierto” y “Falso”. Está compuesto sólo por el conjunto de elementos `True` y `False`, que corresponden a “Cierto” y “Falso”.

¿Recuerdas la selección en los algoritmos? Este tipo es realmente útil dado que se emplea para expresar condiciones que pueden cumplirse o no, y para hacer que los programas ejecuten unas sentencias u otras dependiendo de una condición



dada.

Los operadores disponibles para este tipo son los existentes en el llamado **álgebra de Boole** (de ahí el nombre del tipo).

- Negación: `not`
- Conjunción: `and`
- Disyunción: `or`
- Disyunción exclusiva: `xor`

Sus nombres son pintorescos pero es muy sencillo hacerse con ellos. Sabemos que `True` representa algo que es cierto y `False` representa algo que es falso. Luego si algo no es falso, es que es cierto. Y si algo no es cierto, es que es falso. Esto es:

```
not True = False
```

```
not False = True
```

Sigamos con la conjunción. Si algo que nos dicen es en parte verdad y en parte mentira... ¿Nos están mintiendo? Si queremos ver si dos cosas son conjuntamente verdad utilizamos la conjunción. Sólo es cierta una conjunción de cosas ciertas:

```
False and False = False
```

```
False and True = False
```

```
True and False = False
```

```
True and True = True
```

Sólo resulta ser verdad `True and True` ¡El resto de combinaciones son falsas!

Ahora a por la disyunción. Esta nos dice que alguna de dos cosas es cierta (o tal vez las dos). Esto es:

```
False or False = False
```

```
False or True = True
```

```
True or False = True
```

```
True or True = True
```

Con algún ejemplo más todo esto resultará trivial. Sea `pollofrito` un valor que representa que un pollo está frito y `pollocrudo` un valor que representa que un pollo está crudo. Esto es, si el pollo está frito tendremos que

```
pollofrito = True
```

```
pollocrudo = False
```

Eso sí, cuando el pollo esté crudo lo que ocurrirá será mas bien

```
pollofrito = False
```

```
pollocrudo = True
```

Luego ahora odemos decir que

```
pollofrito or pollocrudo
```

es siempre `True`. Pero eso sí, ¡bajo ningún concepto!, ni por encima del cadáver del pollo, podemos conseguir que

```
pollofrito and pollocrudo
```

sea cierto. Esto va a ser siempre `False`. O está frito o está crudo. Pero hemos quedado

que en este mundo binario nunca vamos a tener ambas cosas a la vez.

Si sólo nos comemos un pollo cuando está frito y el valor `polloingerido` representa que nos hemos comido el pollo, podríamos ver que

```
pollofrito and polloingerido
```

bien podría ser cierto. Pero desde luego

```
pollocrudo and polloingerido
```

tiene mas bien aspecto de ser falso.

El operador booleano que resta por ver no está disponible en todos los lenguajes. Se trata de `xor` (“exclusive or”). Este operador resulta cierto cuando sólo uno de los operandos es cierto. En el resto de casos es falso. Corresponde a la disyunción tal y como la usamos en español (que normalmente excluye uno de los casos cuando el otro es cierto). Estas dos expresiones son equivalentes:

```
A xor B
```

```
(A and (not B)) or ((not A) and B)
```

Los booleanos son extremadamente importantes como estamos empezando a ver. Son lo que nos permite que un programa tome decisiones en función de cómo están las cosas. Por lo tanto tenemos que dominarlos realmente bien para hacer buenos programas.

Tenemos también una gama de operadores que ya conoces de matemáticas que aquí producen como resultado un valor de verdad. Estos son los operadores de comparación, llamados así puesto que se utilizan para comparar valores entre sí:

```
< > <= >= = <>
```

Podemos utilizarlos para comparar valores numéricos, caracteres y booleanos. Pero ambos operandos han de ser del mismo tipo. Estos operadores corresponden a las relaciones *menor que*, *mayor que*, *menor o igual que*, *mayor o igual que*, *igual a* y *distinto a*. Por ejemplo, `3<2` es `False`, pero `2<=2` es `True`.

Por ejemplo, sabemos que si `a` y `b` son dos elementos de tipo `integer`, entonces

```
(a < b) or (a = b) or (a > b)
```

va a ser siempre `True`. No hay ninguna otra posibilidad.

Cuando queramos que un programa tome una decisión adecuada al considerar alguna condición nos vamos a tener que plantear *todos los casos posibles*. Y vamos a tener que aprender a escribir expresiones booleanas para los casos que nos interesen.

Pero cuidado, los números reales se almacenan como aproximaciones. Piénsese por ejemplo en cómo sería posible en otro caso almacenar el número , que tiene infinitas cifras decimales. Por ello no es recomendable comparar números reales empleando “=” o “<>” . Ya veremos más adelante cómo podemos comparar números reales.

La comparación de igualdad y desigualdad puede utilizarse en tipos de datos básicos como los que hemos visto. Las comparaciones que requieren un orden

entre los elementos comparados suelen estar disponibles en tipos de datos numéricos.

Quizá sorprenda que también pueden compararse caracteres. Por ejemplo, esta expresión es cierta, `True`, cuando `x` es un carácter que corresponde a una letra mayúscula:

```
('A' <= x) and (x <= 'Z')
```

Ya comentamos antes que los caracteres son en realidad posiciones en una tabla. Las letras mayúscula están en posiciones consecutivas en dicha tabla de caracteres. Igual sucede con las minúsculas. Los dígitos también. Por eso podemos ver si un carácter corresponde a una letra mayúscula como hemos visto. Pero cuidado, esta expresión no tiene mucho sentido:

```
x <= '0'
```

No sabemos qué caracteres hay antes del carácter para el dígito cero en el código ASCII (salvo si miramos la tabla que describe dicho código). Recuerda que el carácter para el cero no está seguramente en la posición cero de la tabla, y que no tiene nada que ver con el valor cero de tipo entero.

Otro ejemplo más. Debido al tipado estricto de datos en Pascal no podemos comparar

```
3 < '0'
```

dado que no hay **concordancia de tipos** entre los dos valores que comparamos. El primero es de tipo entero, y el segundo de tipo carácter. Esto es, dado que los tipos no son compatibles para usarlos en el mismo operador para compararlos.

## 2.5. Años bisiestos

Podemos combinar todos estos operadores en expresiones más complicadas. Y así lo haremos con frecuencia. Por ejemplo, sea “a” un entero que representa un año, como 2018 o cualquier otro. Esta expresión es `True` si el año `a` es bisiesto:

```
(a mod 4 = 0) and ((a mod 100 <> 0) or (a mod 400 = 0))
```

¿De dónde hemos sacado esto? Los años múltiplos de 4 son bisiestos, excepto los que son múltiplos de 100. Salvo que los múltiplos de 400 lo son. Si queremos una expresión que sea cierta cuando el año es múltiplo de 4 podemos fijarnos en el módulo entre 4. Este será 0, 1, 2 o 3. Para los múltiplos de 4 va a ser siempre 0. Luego

```
a mod 4 = 0
```

es `True` si `a` es múltiplo de 4. Queremos excluir de nuestra expresión aquellos que son múltiplos de 100, puesto que no son bisiestos. Podemos escribir entonces una expresión que diga que *el módulo entre 4 es cero y no es cero el módulo entre 100*. Esto es:

```
(a mod 4 = 0) and not (a mod 100 = 0)
```

Esta expresión es cierta para todos los múltiplos de 4 salvo que sean también múltiplos de 100. Pero esto es igual que

```
(a mod 4 = 0) and (a mod 100 <> 0)
```

que es más sencilla y se entiende mejor. Pero tenemos que hacer que para los múltiplos de 400, a pesar de ser múltiplos de 100, la expresión sea cierta. Dichos años son bisiestos y tenemos que considerarlos. Veamos: suponiendo que *a* es múltiplo de 4, lo que es un candidato a año bisiesto... Si *a* es no múltiplo de 100 o *a* es múltiplo de 400 entonces tenemos un año bisiesto. Esto lo podemos escribir como

```
(a mod 100 <> 0) or (a mod 400 <> 0)
```

pero tenemos naturalmente que exigir que nuestra suposición inicial (que *a* es múltiplo de 4) sea cierta:

```
(a mod 4 = 0) and ((a mod 100 <> 0) or (a mod 400 = 0))
```

¿Cómo hemos procedido para ver si *a* es bisiesto? Lo primero ha sido definir el problema: Queremos **True** cuando *a* es bisiesto y **False** cuando no lo es. Ese es nuestro problema.

Una vez hecho esto, hemos tenido que saber cómo lo haríamos nosotros. En este caso lo mas natural es que no sepamos hacerlo (puesto que normalmente miramos directamente el calendario y no nos preocupamos de calcular estas cosas). Así pues tenemos que aprender a hacerlo nosotros antes siquiera de pensar en programarlo. Tras buscar un poco, aprendemos mirando en algún sitio (¿Google?) que

*“Los años múltiplos de 4 son bisiestos,  
excepto los que son múltiplos de 100.  
Salvo por los múltiplos de 400, que lo son.”*

Nuestro plan es escribir esta definición en Pascal y ver cuál es su valor (cierto o falso). Ahora tenemos que escribir el programa. Lo más importante es en realidad escribir la expresión en Pascal que implementa nuestro plan. Procediendo como hemos visto antes, llegamos a nuestra expresión

```
(a mod 4 = 0) and ((a mod 100 <> 0) or (a mod 400 = 0))
```

¡Ahora hay que probarlo! ¿Será e 2527 un año bisiesto? ¿Y el año 1942? Tomamos prestado el programa para saludar *a* y lo cambiamos para que en lugar de saludar *a* escriba esta expresión. Por el momento no sabemos lo necesario para hacer esto (dado que no hemos visto ningún programa que escriba valores de verdad). En cualquier caso, este sería el programa resultante.

```
1
2 {
3     Es bisiesto con expresiones
4
5 }
6
7 program bisiesto;
8
```

```
9  const
10      Anyo: integer = 1940;
11
12  begin
13      writeln((Anyo mod 4 = 0) and
14              (Anyo mod 100 <> 0) or
15              (Anyo mod 400 = 0));
16  end.
```

Lo que resta es compilar y ejecutar el programa:

```
unix% fpc bisiesto.p
unix% bisiesto
TRUE
```

Cuando se ha compilado el programa, el compilador ha calculado el valor de las expresiones constantes (cuyos operandos son literales, como en todas las de este programa), y ha generado el fichero binario con las instrucciones del programa y los datos que necesita (incluyendo las constantes que hemos definido).

Cuando se ha ejecutado el binario, se ha empezado ejecutando su procedimiento principal. En este caso, el programa principal tiene una única sentencia. Dicha sentencia ha escrito el valor de nuestra expresión (utilizada como argumento `writeln`) como resultado (salida) del programa.

Modifica tú este programa para ver si 1942 es bisiesto o no. Intenta cambiarlo para que un sólo programa te diga si dos años que te gusten son bisiestos.

## 2.6. Más sobre expresiones y constantes

¿Cómo se calcula en Pascal el valor de las constantes? Se calcula el valor correspondiente a las expresiones que sólo utilizan literales y operaciones básicas durante la compilación del programa. Calcular el valor de una expresión se denomina **evaluar** la expresión. Por ejemplo, este programa es otra variante del último ejemplo:

```
1
2  {
3      Es bisiesto con una constante
4
5  }
6
7  program bisiesto;
8
9  const
10      EsBisiesto: boolean = (1900 mod 4 = 0) and
11                          (1900 mod 100 <> 0) or
12                          (1900 mod 400 = 0);
```

```
13
14 begin
15     writeln(EsBisiesto);
16 end.
```

En este caso, el compilador ha calculado un valor de tipo `boolean` para la expresión a partir de la cual hemos definido la constante `EsBisiesto`. La constante queda definida con el valor resultante tras evaluar la expresión. Luego la constante `EsBisiesto` será de tipo `boolean`, como además hemos indicado en la declaración de la constante. Cuando el programa ejecute, siempre que use la constante definida, esta tendrá el mismo valor (el que se calculó cuando se compiló el programa).

Nótese que no todas las expresiones que aparecen en un programa se calculan al compilar el programa. Si la expresión tiene operandos que no son literales, la expresión se evalúa al ejecutar el programa. En realidad, ese será el caso común, como veremos más adelante en el curso. En todo caso, la evaluación de una expresión se realiza de la misma forma en ambos casos.

Veamos otro ejemplo. En este programa, vamos a definir una constante sin indicar siquiera su tipo, para que Pascal sepa que en realidad se puede calcular al compilar como si fuese un literal, y luego utilizaremos dicha constante para definir otra constante. Esta vez, como la primera constante es prácticamente un literal, Pascal no tiene problema en que la utilicemos para definir la siguiente constante.

```
1
2 {
3     Es bisiesto con una constante
4
5 }
6
7 program bisiesto;
8
9 const
10     Anyo = 1900;
11     EsBisiesto: boolean = (Anyo mod 4 = 0) and
12         (Anyo mod 100 <> 0) or
13         (Anyo mod 400 = 0);
14
15 begin
16     writeln(EsBisiesto);
17 end.
```

Nosotros vamos a utilizar siempre la declaración de constantes que indica el tipo de datos de la constante, lo que nos va a ayudar a detectar errores lo antes posible. Sólo

cuando no quede otro remedio, definiremos constantes como en este ejemplo sin indicar su tipo (para poderlas utilizar en otras definiciones). En ambos casos, no hay ningún problema en utilizarlas en sentencias, que es lo que importa.

## 2.7. Evaluación de expresiones

Las expresiones se evalúan siempre *de dentro hacia afuera*, y normalmente *de izquierda a derecha* (aunque no es bueno depender de esto último en general). Esto quiere decir que las expresiones se evalúan haciendo caso de los paréntesis (y de la precedencia de los operadores) de tal forma que primero se hacen los cálculos más interiores o **anidados** en la expresión. Por ejemplo, si partimos de

```
(A mod 4 = 0) and ((A mod 100 <> 0) or (A mod 400 = 0))
```

y `A` tiene como valor `444`, entonces Pascal evalúa

```
(444 mod 4 = 0) and ((444 mod 100 <> 0) or (444 mod 400 = 0))
```

Aquí, Pascal va a calcular primero `444 mod 4`. Luego podría calcular `444 mod 100` y luego `444 mod 400`. Estas sub-expresiones son las más internas o más anidadas de la expresión. Supongamos que tomamos la primera. Esto nos deja:

```
0 = 0 and ((444 mod 100 <> 0) or (444 mod 400 = 0))
```

Pascal seguiría de dentro hacia afuera y de izquierda a derecha:

```
0 = 0 and ((444 mod 100 <> 0) or (444 mod 400 = 0))
```

```
True and ((444 mod 100 <> 0) or (444 mod 400 = 0))
```

```
True and ((44 <> 0) or (444 mod 400 = 0))
```

```
True and (True or (444 mod 400 = 0))
```

En este punto pasa algo curioso (en este dialecto de Pascal y en algunos lenguajes, pero no en todos). Como `True or` cualquier cosa es simplemente `True`, Pascal se evita calcular el segundo operando de la operación `or` y nos queda:

```
True and True
```

Lo que por supuesto es

```
True
```

A esto, dejar de evaluar expresiones booleanas en cuanto se sabe que son ya `True` o `False`, se lo denomina **evaluación con cortocircuito** de expresiones booleanas. Hay lenguajes que seguirían evaluando

```
True and (True or (444 mod 400 = 0))
```

```
True and (True or (44 = 0))
```

```
True and (True or False)
```

```
True and True
```

```
True
```


Pero con cortocircuito tenemos:

```
True and (True or (444 mod 400 = 0))
```

```
True and True
```

```
True
```

Si no tenemos paréntesis recurrimos a la precedencia para ver en qué orden se evalúan las cosas. En la figura 2 mostramos todos los operadores, de mayor a menor precedencia (los que están en la misma fila tienen la misma precedencia). En Pascal, los operadores con la misma precedencia se evalúan de izquierda a derecha (ten cuidado, esto no es así en todos los lenguajes de programación).



**Fig. 2.** Precedencia de operadores en Pascal.

Si tomamos por ejemplo la expresión

```
not False and False or True
```

podemos ver que `not` va antes que `and`, por lo que en realidad tenemos

```
((not False) and (False or True))
```

Para evaluarla vamos de izquierda a derecha y de dentro de los paréntesis hacia afuera. Por ejemplo:

```
((not False) and (False or True))
```

```
(True and (False or True))
```

```
(True and True)
```

```
True
```

Por cierto, hay dos leyes muy útiles en expresiones booleanas que son las denominadas **leyes de De Morgan** y que podemos utilizar para simplificar expresiones booleanas. Según estas leyes, resulta que las siguientes expresiones son equivalentes:

```
not (a and b)
```

```
(not a) or (not b)
```

Y las siguientes son también equivalentes:

```
not (a or b)
```

```
(not a) and (not b)
```

Puedes comprobarlo. Sugerimos intentar eliminar los `not` de las condiciones en los



programas, puesto que los humanos entienden peor las negaciones que las afirmaciones ¿No te parece que no sucede que no es cierto? O tal vez deberíamos decir... ¿No es cierto?

Un último detalle importante sobre las expresiones. Como ya se explicó en el capítulo anterior, el compilador hace todo lo posible por detectar fallos en el programa. Algunos errores en expresiones se detectan al compilar el programa, generando un error de compilación.

El compilador evalúa las expresiones cuyos operandos son literales, ya que sus valores se conocen cuando el compilador traduce el programa. Gracias a esto, el compilador es capaz de detectar ciertos errores en las expresiones evitando que el programa tenga errores de ejecución. Por ejemplo, si en una expresión se está dividiendo por cero, el compilador se negará a generar el fichero binario y dará errores de compilación. Naturalmente, si la evaluación sucede cuando ejecute el programa (en tiempo de ejecución) y es entonces cuando dividimos por cero, el programa tendrá un error durante su ejecución.

Pero atención, esto no nos libra de tener otros errores lógicos o de ejecución en nuestros programas.

## 2.8. Elementos predefinidos en Pascal

En Pascal existen diversas funciones, operaciones y constantes predefinidas en el lenguaje (en inglés, *built-in*) que resultan muy útiles para construir expresiones. Estas funciones y constantes pueden ayudarnos a obtener valores que dependen de un tipo de datos determinado.

Los tipos ordinales, esto es, los tipos de datos cuyos valores se pueden contar y tienen un orden, tienen definidas las funciones

`pred(v)`

`succ(v)`

que, respectivamente, evalúan al valor predecesor (anterior) y sucesor (posterior) al valor *v* (en el tipo de datos correspondiente). Por ejemplo, si *c* es de tipo `char` con valor

`'B'`

entonces

`pred(c)`

tiene como valor

`'A'`

Siguiendo con el mismo ejemplo,

`succ(c)`

tiene como valor el siguiente valor en el tipo de datos `char`, que es

`'C'`

Las funciones predefinidas en nuestro dialecto de Pascal son las siguientes. En cada una utilizamos “*i*” como nombre del argumento si la función existe sólo para

números enteros, “n” si existe para números (enteros o reales), y “c” si existe para caracteres (o tipos ordinales en general). Utilizamos “x” si la función existe para cualquier ordinal o array (ya veremos qué es esto).

Built-in	Función
abs(n)	valor absoluto
arctan(n)	arcotangente
cos(n)	coseno
exp(n)	exponencial
ln(n)	logaritmo neperiano
round(n)	redondear a entero
sin(n)	seno
sqr(n)	elevar al cuadrado
sqrt(n)	raíz cuadrada
trunc(n)	truncar a entero
chr(i)	carácter en posición i.
ord(c)	posición del carácter o valor c.
pred(c)	carácter o valor predecesor de c.
succ(c)	carácter o valor sucesor de c.
low(x)	menor valor o índice en x.
high(x)	mayor valor o índice en x.

Un procedimiento *built-in* llamado `halt` nos permite abortar la ejecución del programa en cualquier punto. Esto sólo se debe hacer cuando nuestro programa ha detectado una situación de la que no se puede recuperar. Ya veremos ejemplos durante el curso. Por ejemplo, si el programa llega a ejecutar la sentencia

```
halt();
```

entonces se acaba su ejecución en el acto. A medida que avancemos, irán apareciendo nuevos *built-ins*.

Como ejemplo, estas dos sentencias escribirían por la pantalla el máximo entero y su predecesor:

```
writeln(high(integer));
```

```
writeln(pred(high(integer)));
```

No podemos usar los nombres de las funciones y las constantes predefinidas como identificadores en nuestros programas, ya que esos nombres están reservados para el propio lenguaje (como las palabras reservadas).

## 2.9. Longitud de una circunferencia

Vamos a poner todo esto en práctica. Queremos un programa que escriba la longitud de una circunferencia de radio  $r$ . El problema es: dado  $r$ , calcular  $2\pi r$ , naturalmente. Podríamos modificar el último programa que hemos hecho para conseguir esto. Pero vamos a escribirlo desde el principio. Lo primero que necesitamos es escribir la estructura de nuestro programa. Podemos empezar por...

```
1
2 {
3     caular longitud de circunferencia de radio r
4 }
5 program longcircunf;
6 begin
7     writeln(0);
8 end.
```

No es que haga mucho. De hecho, hemos utilizado la sentencia nula como cuerpo del programa principal: dejando únicamente el punto y coma. Esto quiere decir que Pascal no va a ejecutar nada cuando ejecute el cuerpo del programa. Ahora compilamos y ejecutamos el programa para ver al menos que está bien escrito. Evidentemente, no veremos nada en la pantalla al ejecutar el programa, porque no hace nada.

Lo siguiente que necesitamos es nuestra constante  $\pi$ . Podemos utilizar la que teníamos escrita en otros programas y ponerla al principio en la zona de declaraciones de constantes.

```
1
2 {
3     caular longitud de circunferencia de radio r
4 }
5 program longcircunf;
6
7 const
8     Pi: real = 3.1415926;
9 begin
10     writeln(0);
11 end.
```

De nuevo compilamos y ejecutamos el programa. Al menos sabremos si tiene errores sintácticos o no. Si los tiene será mucho más fácil encontrar los errores ahora que luego.

Ahora podríamos definir otra constante para el radio que nos interesa para el cálculo, `radio`, y por último escribir la expresión en la sentencia `writeln`. El programa

resultante es como sigue:

```
1
2 {
3     caular longitud de circunferencia de radio r
4 }
5 program longcircunf;
6
7 const
8     Pi: real = 3.1415926;
9     R: real = 3.0;
10 begin
11     writeln(2*Pi*R);
12 end.
```

Si lo compilamos de nuevo y lo ejecutamos tenemos el resultado.

```
unix% fpc longcircunf.p
unix% longcircunf
2.8274333400000001E+002
```

Este programa está bien. Pero lo suyo es que ya que la longitud de una circunferencia es en realidad una función, el programa utilice una función para calcularla (como el programa que utilizamos para ver la anatomía de un programa al principio del curso). Podemos copiar la función que aparecía en ese programa y adaptarla a nuestros propósitos. sta nos quedaría como vemos en este programa.

```
1
2 {
3     caular longitud de circunferencia de radio r
4 }
5 program longcircunf;
6
7 const
8     Pi: real = 3.1415926;
9
10 function longitudcircunferencia(r: real): real;
11 begin
12     result := 2*Pi*r;
13 end;
14
15 const
16     Radio: real = 3.0;
17 begin
18     writeln(longitudcircunferencia(Radio));
19 end.
```

La sintaxis puede ser rara, pero es fácil de entender. La línea 10 define una función llamada `longitudcircunferencia` a la que podemos dar un número de tipo `real` como argumento para que nos devuelva otro número de tipo `real`. El número que le damos a la función lo llamamos `r` en la definición de la función. Esto es, `r` es el parámetro de la función. Esta línea (la 10) es la que llamamos cabecera de la función. Entre “`begin`” y “`end`” tenemos que escribir cómo se calcula el valor de la función. A esto lo llamamos cuerpo de la función. La sentencia que tenemos hace que la función devuelva el valor de la expresión escrita a la derecha de “`:=`” cada vez que la utilicemos. En el cuerpo del programa principal escribimos el valor que devuelve la función para un radio de prueba que hemos definido antes. En

```
write(longitudcircunferencia(Radio));
```

La expresión

```
longitudcircunferencia(Radio)
```

tiene como valor (calculará) la longitud de la circunferencia de radio `Radio` que en este caso es 3.0.

En adelante utilizaremos funciones para realizar nuestros cálculos, como en este último programa.

## 2.10. Problemas

1. Modifica el último programa que hemos mostrado para que calcule el área de un círculo de radio dado en lugar de la longitud de la circunferencia dado el radio. *Presta atención a las normas de estilo.* Esto quiere decir que tendrás que cambiar el nombre del programa y tal vez otras cosas además de simplemente cambiar el cálculo.
2. Escribe en Pascal las siguientes expresiones. Recuerda que para que todos los números utilizados sean números reales has de poner siempre su parte decimal, aunque ésta sea cero. En algunos casos te hará falta utilizar funciones numéricas predefinidas en Pascal, tales como la raíz cuadrada. Consulta el libro para encontrar las funciones necesarias.

A)

$$2 \cdot 7^2 + -3 \cdot 2^2$$

B) Para al menos 3 valores de  $r$ ,

$$\frac{4}{3} r^3$$

C) El factorial de 5, que suele escribirse como 5!. El factorial de un número natural es dicho número multiplicado por todos los naturales menores que el hasta el 1. (Por definición, 0! se supone que tiene como valor 1).

D) Dados  $a = 4$  y  $b = 3$

$$\binom{a}{b} = \frac{a!}{b!(a-b)!}$$

E)

$$\sqrt{\pi}$$

F) Para varios valores de  $x$  entre 0 y 2,

$$\sin^2 x + \cos^2 x$$

G) Siendo  $x = 2$ ,

$$\frac{1}{\sqrt{3.5x^2 + 4.7x + 9.3}}$$

3. Por cada una de las expresiones anteriores escribe un programa en Pascal que imprima su valor.

A) Hazlo primero declarando una constante de prueba que tenga como valor el de la expresión a calcular.

B) Hazlo ahora utilizando una función para calcular la expresión.

4. Para cada una de las expresiones anteriores indica cómo se evalúan éstas, escribiendo cómo queda la expresión tras cada paso elemental de evaluación hasta la obtención de un único valor como resultado final.

# Índice

Elementos básicos .....	1
2.1. ¿Por dónde empezamos? .....	1
2.2. Operaciones .....	4
2.3. Expresiones .....	5
2.4. Otros tipos de datos .....	6
2.5. Años bisiestos .....	10
2.6. Más sobre expresiones y constantes .....	12
2.7. Evaluación de expresiones .....	14
2.8. Elementos predefinidos en Pascal .....	16
2.9. Longitud de una circunferencia .....	18
2.10. Problemas .....	20